


```

cccccccc
      subroutine rk4(n,x,y,xend)
c  RK4 integrates a single step, from x to xend, of the system of n
c  first-order differential equations dy/dx=yprime(x,y) given by
c  subroutine fcn, which must be provided by the user.
c  On input y contains y(x); x=xend and y=y(xend) are returned.
c  N.B. this subroutine makes no attempt to monitor or control errors!
c
      implicit double precision (a-h,o-z)
      dimension y(n),yprime(10),y1(10),y2(10)
c
      if (n.gt.10) stop 'Storage exceeded in subroutine rk4!'
      delx=xend-x
      delx2=0.5d0*delx
c  Stage 1 of RK4.
      call fcn(n,x,y,yprime)
      do 10 i=1,n
      y1(i)=delx2*yprime(i)
      y(i)=y(i)+y1(i)
10    continue
c  Stage 2 of RK4.
      x=x+delx2
      call fcn(n,x,y,yprime)
      do 20 i=1,n
      y2(i)=delx2*yprime(i)
      y(i)=y(i)+y2(i)-y1(i)
20    continue
c  Stage 3 of RK4.
      call fcn(n,x,y,yprime)
      do 30 i=1,n
      y1(i)=y1(i)/3.0d0
      y2(i)=delx*yprime(i)-y2(i)
      y(i)=y(i)+y2(i)
30    continue
c  Stage 4 of RK4.
      x=x+delx2
      call fcn(n,x,y,yprime)
      do 40 i=1,n
      y1(i)=y1(i)-y2(i)
      y2(i)=delx2*yprime(i)+y2(i)
      y(i)=y(i)+y1(i)+y2(i)/3.0d0
40    continue
c
      return
      end

```